
Abstract

In this paper we show how an arbitrary Turing machine can be simulated in DATR,

and show that the computational complexity of DATR is Turing equivalent { and hence

termination of query evaluation is undecidable.

1 Introduction

(Moser 1992b) proved a lower bound of co-NP-hard for the computational complexity of DATR

(Evans and Gazdar 1989a, Evans and Gazdar 1989b, Evans and Gazdar 1990) query evaluation,

�rst by showing how an NP-complete problem { determining satis�ability of an arbitrary boolean

formula { could be solved in DATR. It then showed that the complement of the set of satis�able

boolean formulas of propositional calculus, those which are unsatis�able, can also be recognised

by a DATR theory.

To say that the computational complexity is undecidable means that it is impossible to estab-

lish a lower bound on the time taken for a well-formed, functional theory to evaluate a query. Such

a theory may recognise a language for which there is no e�ective algorithm (one which halts) for

recognising non-membership in the language. Recognising membership of a string in the language

will halt, but one cannot predict in general whether the string being tested is in the language or

not.

2 The Hopcroft & Ullman Turing machine

Following Hopcroft & Ullman (1979, pp. 147-150) we de�ne a Turing machine as an ordered tuple

M = (Q;�;�; �; q

0

; B; F )

where

Q is the �nite set of states,

� is the �nite set of allowable tape symbols,

B, a symbol of �, is the blank,

�, a subset of � not including B, is the set of input symbols,

� is the next move function, a partial function � : Q� �! Q� �� fL;Rg

q

0

2 Q is the start state,

F � Q is the set of �nal states.

An instantaneous description (ID) of the Turing machine M is denoted by �

1

q�

2

, where q is

the current state, and �

1

�

2

, a string in �

�

is the current content of the tape up to the rightmost

nonblank symbol. The tape is bounded on the left (at the leftmost symbol of �

1

) and is unbounded

on the right, all cells to the right of �

2

having value B. The tape head is scanning the leftmost

symbol of �

2

, unless �

2

is the empty string, in which case the head is scanning a blank.

A Turing machine which recognises the language L = f0

n

1

n

j n � 1g is given in Hopcroft &

Ullman. The TM M is de�ned by:

Q = fq

0

; q

1

; q

2

; q

3

; q

4

g;

� = f0; 1g;

� = f0; 1; X; Y;Bg;

F = fq

4

g;

and the transition function � is as de�ned in Figure 1.

Symbol

State 0 1 X Y B

q

0

(q

1

; X;R) � � (q

3

; Y;R) �

q

1

(q

1

; 0; R) (q

2

; Y; L) � (q

3

; Y;R) �

q

2

(q

0

; 0; L) � (q

0

; X;R) (q

2

; Y; L) �

q

3

� � � (q

3

; Y;R) (q

4

; B;R)

q

4

� � � � �

Figure 1: The function �

Figure 2 shows the computation of M accepting string 0011:

q

0

0011 ` Xq

1

011 ` X0q

1

11 ` Xq

2

0Y 1 `

q

2

X0Y 1 ` Xq

0

0Y 1 ` XXq

1

Y 1 ` XXY q

1

1 `

XXq

2

Y Y ` Xq

2

XY Y ` XXq

0

Y Y ` XXY q

3

Y `

XXY Y q

3

` XXY Y Bq

4

Figure 2: A computation of M

3 A DATR representation

To simulate a Turing machine we require representations of an in�nite tape, a �nite-state control,

and each of the elements of the ordered tuple M . The simulated Turing machine uses the path

to represent the tape, or more precisely as a path pre�x; and the control we represent as a

combination of a path pre�x and an inheritance speci�cation. An ID �

1

q�

2

is represented as a

three-argument list:

<x

0

x

1

x

2

: : : x

i�1

; q ; x

i

x

i+1

: : : x

n

;>

where �

1

= x

0

x

1

x

2

: : : x

i�1

, �

2

= x

i

x

i+1

: : : x

n

, and x

i

is the current symbol. Using tech-

niques presented in Moser (1992a) to treat the path as an argument list, we will access the path

as a stack accessed from the left side. We de�ne a main node, say M, such that the value of a

path (representing an ID) at that node is the value which Turing machine M would compute if

started from that ID. We do this using two mutually recursive nodes, M and Apply delta. A step

of computation from ID

i

to ID

i+1

requires two cycles through M and one through Apply delta.

The �rst cycle through M computes the value of �(q; 
) (a triple) and pushes it (as a path pre�x),

prefaced by the atom delta. The second cycle through M tests whether � returned a value or is

unde�ned. If it is not de�ned, then M evaluates to either accept or reject depending on whether

q is a �nal state. If �(q; 
) is de�ned, then M inherits from Apply delta, which pops �(q; 
) (as

a matched pre�x), applies it to the current ID

i

, and inherits from M:<ID

i+1

>. Figure 3 outlines

the mutual recursion through which the computation of M is simulated, where the last step of

computation could be either accept, as shown, or reject.

M:<ID

1

> = M:<delta �(q; 
) ID

1

>

= Apply delta:<�(q; 
) ID

1

>

= M:<ID

2

>

= M:<delta �(q; 
) ID

2

>

= Apply delta:<�(q; 
) ID

2

>

= M:<ID

3

>

.

.

.

.

.

.

= M:<ID

n

>

= M:<delta undef ID

n

>

= accept

Figure 3: Computation via mutual recursion of M and Apply delta

We now present the DATR translation of TM M = (Q;�;�; �; q

0

; B; F ). We represent states (Q)

and tape symbols (�) as atoms, and de�ne DATR variables ranging over each of these sets.

% states

#vars $state: q0 q1 q2 q3 q4.

% tape symbols

#vars $gamma: 0 1 x y b.

The de�nition of node M will use argument manipulation primitives de�ned in Moser (1992a).

These primitives are parameterized over several variables, the most interesting being $terminal,

which enumerates the atoms over which arguments are constructed. The two separators ; and

! serve to terminate arguments and the argument list, repsectively. For a Turing machine, the

$terminals are the state and tape symbols, plus moves L and R:

#vars $terminal: q0 q1 q2 q3 q4 0 1 x y b l r.

Under the path-as-argument-list interpretation, �

1

, q, and �

2

are the �rst, second and third

arguments, respectively, so we de�ne several nodes which function as argument extractors:

Alpha1:<> == Arg1.

Curq:<> == Arg2.

Alpha2:<> == Arg3.

The transition function � is simply a table look-up. �(q; 
) = (q

0

; 


0

; d), or in our DATR notation

Delta:<q g> == (q

0

; g

0

; d), where q and g are the current state and tape symbol being

scanned, q

0

, g

0

, and d are the new state, the symbol replacing g on the tape, and the direction

in which the head moves, respectively. Noting that this particular transition table is a sparse

matrix, we de�ne a default of undef and the specify the value of � for the pairs for which it is

de�ned:

Delta: <> == undef

<q0 0> == (q1 ; x ; r ;)

<q0 y> == (q3 ; y ; r ;)

<q1 0> == (q1 ; 0 ; r ;)

<q1 1> == (q2 ; y ; l ;)

<q1 y> == (q1 ; y ; r ;)

<q2 0> == (q2 ; 0 ; l ;)

<q2 x> == (q0 ; x ; r ;)

<q2 y> == (q2 ; y ; l ;)

<q3 y> == (q3 ; y ; r ;)

<q3 b> == (q4 ; b ; r ;).

Testing membership in the set of �nal states requires one non-default statement for each node in

F , as shown in node Final:

1

Final: <> == false

<q4> == true.

The current symbol scanned by the read/write head of M is the leftmost symbol of �

2

, unless

�

2

is nil, in which case the current symbol is the blank. Node Cursym evaluates to the current

symbol using negative path extension: the statement pre�xed by <nil ;> will be matched when

the value of Alpha2 is the empty list; otherwise the statement pre�xed by <nil> will be matched:

Cursym:<> == <nil Alpha2 ;>

<nil ;> == b

<nil> == First:<>.

The e�ect of evaluating an ID at Cursym yields the �rst symbol of �

2

. In the second theorem

below, �

2

is the empty string:

Cursym: <; q0 ; 0 0 1 1 ;> = 0

<x x y y ; q3 ; ;> = b.

Before presenting the de�nition of M we �rst introduce a new primitive, Last, which evaluates

to the last atom in an argument, or the empty list if the argument is nil. This will be used to

simulate moving the read/write head to the left; the rightmost symbol of �

1

needs to be removed

from �

1

and inserted to the right of the current state q.

1

Moser (1992c) discusses the representation of disjunction in DATR at length.

Last:<$terminal ;> == $terminal

<;> == ()

<$terminal> == <>.

We now present the de�nition of M such that M:<ID> = accept, or M:<ID> = reject, where ID

is of the form <x

0

x

1

x

2

: : : x

i�1

; q ; x

i

x

i+1

: : : x

n

;>:

M:<> == <delta Delta:<Curq Cursym>>

<delta undef> == <If:<Final:<Curq:<>> > >

<then> == accept

<else> == reject

<delta> == Apply_delta:<>.

The �rst statement of M indicates that

M:<ID

i

> = M:<delta �(q; 
) ID

i

>

and from the last statement of M we see that

M:<delta �(q; 
) ID

i

> = Apply delta:<�(q; 
) ID

i

>

and we require that

Apply delta:<�(q; 
) ID

i

> = M:<ID

i+1

>

For each triple (q

0

; 


0

; d) in the range of �, node Apply delta has a path de�nition which con-

structs ID

i+1

from the ID

i

su�xed to the triple. Node Apply delta has two forms of statement,

corresponding to transitions which move the read/write head left and right. For example, for

(q

1

; X;R) (= �(q

0

; 0)) Apply delta has the following statement:

Apply_delta: <q1 ; x ; r ;> == M:<Alpha1:<> x ; q1 ; Rest:<Alpha2:<> ;> ; !>.

From this statement (and the de�nitions of the nodes on which it depends), it follows that:

Apply delta:<q1 ; x ; r ; x

0

x

1

x

2

: : : x

i�1

; q0 ; x

i

x

i+1

: : : x

n

;>

= M:<x

0

x

1

x

2

: : : x

i�1

x ; q1 ; x

i+1

: : : x

n

; !>

When the read/write head moves left, as in (q

2

; Y; L) (= �(q

2

; Y )) Apply delta has the following

statement:

Apply_delta: <q2 ; y ; l ;> == M:< Remove_last:<Alpha1:<> ;> ;

q2 ;

Last:<Alpha1:<> ;> y Rest:<Alpha2:<> ;> ;

!>.

From this statement (and the de�nitions of the nodes on which it depends), it follows that:

Apply delta:<q2 ; y ; l ; x

0

x

1

x

2

: : : x

i�1

; q0 ; x

i

x

i+1

: : : x

n

;>

= M:<x

0

x

1

x

2

: : : x

i�2

; q2 ; x

i�1

y x

i+1

: : : x

n

; !>

Although Apply delta need have only the same number of statements as there are triples mapped

onto by � (in this case 10), it is simpler and clearer to de�ne a statement for every triple in

Q� �� fL;Rg, as illustrated below using the DATR variables de�ned previously:

Apply_delta:

<$state ; $gamma ; r ;> == M:< Alpha1:<> $gamma ;

$state ;

Rest:<Alpha2:<> ;> ;

!>

<$state ; $gamma ; l ;> == M:< Remove_last:<Alpha1:<> ;> ;

$state ;

Last:<Alpha1:<> ;> $gamma Rest:<Alpha2:<> ;> ;

!>.

The number of statements comprising node Apply delta is jQj�j�j�jfL;Rgj, since the variables

get expanded out to one statement for each atom in their respective ranges.

The �rst four IDs in the computation of M:<; q0 ; 0 0 1 1 ;> are shown in Figure 4, along

with the intermediate stages. A comparison with Figure 2 shows that the sequence of IDs is

identical to those progressed through by Turing machineM .

M:<; q0 ; 0 0 1 1 ;>

= M:<delta q1 ; x ; r ; ; q0 ; 0 0 1 1 ;>

= Apply delta:<q1 ; x ; r ; ; q0 ; 0 0 1 1 ;>

= M:<x ; q1 ; 0 1 1 ; ! ; q0 ; 0 0 1 1 ;>

= M:<delta q1 ; x ; r ; ; q0 ; 0 0 1 1 ;>

= Apply delta:<q1 ; 0 ; r ; x ; q1 ; 0 1 1 ; ! ; q0 ; 0 0 1 1 ;>

= M:<x 0 ; q1 ; 1 1 ; ! x ; q1 ; 0 1 1 ; ! ; q0 ; 0 0 1 1 ;>

= M:<delta q2 ; y ; l ; x 0 ; q1 ; 1 1 ; ! x ; q1 ; 0 1 1 ; ! ; q0 ; 0 0 1 1 ;>

= Apply delta:<q2 ; y ; l ; x 0 ; q1 ; 1 1 ; ! x ; q1 ; 0 1 1 ; ! ; q0 ; 0 0 1 1 ;>

= M:<x ; q2 ; 0 y 1 ; ! x 0 ; q1 ; 1 1 ; ! x ; q1 ; 0 1 1 ; ! ; q0 ; 0 0 1 1 ;>

Figure 4: The �rst few steps of computation

When Apply delta inherits from M, the default extension which is appended to the constructed

path is the previous sequence of IDs. After n steps of computation the path looks like:

< ID

n

! : : : ! ID

3

! ID

2

! ID

1

>

To prevent the trailing sequence of IDs from `interfering' with the simulated computation, the

contructed path is terminated with !, a symbol which the argument extractors (Arg1, Arg2, etc.)

never look beyond.

The last computation of M occurs when � is unde�ned. In this case the pre�x inserted on the

�rst cycle is delta undef and on the second cycle evaluation terminates. Figure 5 (page 7) shows

the last step of computation for the sequence begun in Figure 4 case where M accepts. Theorems

of this theory conform to the results computed by the equivalent Turing machine. By starting

compution at the initial ID we have:

M: <; q0 ; 0 ;> = reject

<; q0 ; 1 ;> = reject

<; q0 ; 0 1 ;> = accept

<; q0 ; 0 0 1 ;> = reject

<; q0 ; 0 1 1 ;> = reject

<; q0 ; 0 0 1 1 ;> = accept

<; q0 ; 0 0 0 1 1 ;> = reject

<; q0 ; 0 0 0 1 1 1 ;> = accept.

4 Conclusion

The mapping from the statement of the Turing machine M to a DATR theory which executes

it is straightforward. The size of the DATR theory is linear with respect to the size of the

Turing machine. The language accepted by the DATR theory is identical to that accepted by the

Turing machine, and the abstract steps of computation are identical. In practice, however, the

= M:< x x y y b ; q4 ; ; !

x x y y ; q3 ; ; ! x x y ; q3 ; y ; ! x x ; q0 ; y y ; !

x x y y ; q3 ; ; ! x x y ; q3 ; y ; ! x x ; q0 ; y y ; !

x ; q2 ; x y y ; ! x x ; q2 ; y y ; ! x x y ; q1 ; 1 ; !

x x ; q1 ; y 1 ; ! x ; q0 ; 0 y 1 ; ! ; q2 ; x 0 y 1 ; !

x ; q2 ; 0 y 1 ; ! x 0 ; q1 ; 1 1 ; ! x ; q1 ; 0 1 1 ; !

; q0 ; 0 0 1 1 ;>

= M:< delta undef x x y y b ; q4 ; ; !

x x y y ; q3 ; ; ! x x y ; q3 ; y ; ! x x ; q0 ; y y ; !

x ; q2 ; x y y ; ! x x ; q2 ; y y ; ! x x y ; q1 ; 1 ; !

x x ; q1 ; y 1 ; ! x ; q0 ; 0 y 1 ; ! ; q2 ; x 0 y 1 ; !

x ; q2 ; 0 y 1 ; ! x 0 ; q1 ; 1 1 ; ! x ; q1 ; 0 1 1 ; !

; q0 ; 0 0 1 1 ;>

= M:< then x x y y b ; q4 ; ; !

x x y y ; q3 ; ; ! x x y ; q3 ; y ; ! x x ; q0 ; y y ; !

x ; q2 ; x y y ; ! x x ; q2 ; y y ; ! x x y ; q1 ; 1 ; !

x x ; q1 ; y 1 ; ! x ; q0 ; 0 y 1 ; ! ; q2 ; x 0 y 1 ; !

x ; q2 ; 0 y 1 ; ! x 0 ; q1 ; 1 1 ; ! x ; q1 ; 0 1 1 ; !

; q0 ; 0 0 1 1 ;>

= accept.

Figure 5: The �nal step of computation by node M

DATR theory does more work in the parameter passing, as the default extension consisting of

the sequence of IDs comprising the computation is appended repeatedly to the path passed from

Apply delta to M.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %

% File: turing.dtr %

% Purpose: Simulate a Turing machine. %

% Author: Lionel Moser, June 1992 %

% Documentation: HELP *datr %

% Related Files: lib datr; args.dtr; arglogic.dtr; %

% Version: 2.00 %

% Copyright (c) University of Sussex 1992. All rights reserved. %

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %

% This theory simulates a Turing machine which recognises the language

% L={0^n 1^n | n >= 1}, taken from Hopcroft & Ullman (1979:147-150).

% The DATR theory simultates a TM defined by the ordered tuple

% M=(Q,Sigma,Gamma,Delta,q0,B,F).

#vars $state: q0 q1 q2 q3 q4. % Q

#vars $move: l r. % {L,R}

% Tape symbols: 0, 1, x, y, b % B = b

#vars $gamma: 0 1 x y b. % Gamma

#vars $sigma: 0 1. % Sigma

#vars $final: q4. % F

#vars $terminal: q0 q1 q2 q3 q4 l r 0 1 x y b.

#load 'args.dtr'. % The required extract of these

#load 'arglogic.dtr'. % files follows in primitives.dtr

% An instantaneous description (ID) is a string

% <Alpha1 q Alpha2>

% where Alpha1 and Alpha2 are strings over the tape alphabet.

% We represent this as a 3-argument list:

% <X0 X1 X2 ... Xi-1 ; q ; Xi ... Xn ;>

Alpha1:<> == Arg1. % X0 X1 ... Xi-1

Curq:<> == Arg2. % current state q

Alpha2:<> == Arg3. % Xi Xi+1 ... Xn

% Current symbol is Xi, or b (blank) if Alpha2 is the empty string.

Cursym:<> == <nil Alpha2 ;>

<nil ;> == b

<nil> == First:<>.

% Final states (= {q4} in this example)

Final: <> == false

<$final> == true.

% The Delta function (a partial function) is stored as a look-up table.

% Delta:<q a> == (q' ; a' ; {r/l} ;)

Delta:

<> == undef

<q0 0> == (q1 ; x ; r ;)

<q0 y> == (q3 ; y ; r ;)

<q1 0> == (q1 ; 0 ; r ;)

<q1 1> == (q2 ; y ; l ;)

<q1 y> == (q1 ; y ; r ;)

<q2 0> == (q2 ; 0 ; l ;)

<q2 x> == (q0 ; x ; r ;)

<q2 y> == (q2 ; y ; l ;)

<q3 y> == (q3 ; y ; r ;)

<q3 b> == (q4 ; b ; r ;).

% Last is the last symbol in an argument, or nil if the arg is nil.

Last:<$terminal ;> == $terminal

<;> == ()

<$terminal> == <>.

% M:<ID>

% M:<X0 ... Xi-1 ; q ; Xi Xi+1 ... Xn ;>

M:<> == <delta Delta:<Curq Cursym>>

<delta undef> == <If:<Final:<Curq:<>> > >

<then> == accept

<else> == reject

<delta> == Apply_delta:<>.

% Apply_delta:<Delta(IDi) IDi> == M:<IDi+1>

% Apply_delta<q1 ; X ; R ; X0 ... Xi-1 ; q ; Xi Xi+1 ... Xn ;>

Apply_delta:

% M:<X0 ... Xi-1 Xi' ; q' ; Xi+1 ... Xn ;>

<$state ; $gamma ; r ;> == M:< Alpha1:<> $gamma ;

$state ;

Rest:<Alpha2:<> ;> ;

!>

% M:<X0 ... Xi-2 ; q' ; Xi Xi' Xi+1 ... Xn ;>

<$state ; $gamma ; l ;> == M:< Remove_last:<Alpha1:<> ;> ;

$state ;

Last:<Alpha1:<> ;> $gamma Rest:<Alpha2:<> ;> ;

!>.

% Some theorems ----------------------

% M: <; q0 ; 0 ;> = reject

% <; q0 ; 1 ;> = reject

% <; q0 ; 0 1 ;> = accept

% <; q0 ; 0 0 1 ;> = reject

% <; q0 ; 0 1 1 ;> = reject

% <; q0 ; 0 0 1 1 ;> = accept

% <; q0 ; 0 0 0 1 1 ;> = reject

% <; q0 ; 0 0 0 1 1 1 ;> = accept

% <; q0 ; 0 0 0 1 1 1 1 ;> = reject.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %

% %

% File: primitives.dtr %

% Purpose: Primitives used by Turing.dtr %

% Authors: Lionel Moser, June 1992. %

% Version: 5.00 %

% Copyright (c) University of Sussex 1992. All rights reserved. %

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % %

Arg1: <!> == ()

<;> == ()

<$terminal> == ($terminal <>).

Arg2: <> == Arg1:<Pop_arg>.

Arg3: <> == Arg1:<Pop_arg:<Pop_arg>>.

First: <$terminal> == $terminal.

Second: <$terminal> == First:<>.

Pop_arg: <!> == ()

<;> == Arglist:<>

<$terminal> == <>.

Pv: <> == ()

<!> == ()

<;> == (; <>)

<$terminal> == ($terminal <>).

Arglist:<> == Pv.

Rest: <;> == ()

<$terminal> == Pv_to_;:<>.

Pv_to_;: <;> == ()

<$terminal> == ($terminal <>).

Remove_last: <> == Reverse:<Rest:<Reverse ;> ;>.

Reverse: <;> == ()

<$terminal> == (<> $terminal).

If: <true> == then

<false> == else.

